<?xml version='1.0'?>
<!--=======================================================================
	Copyright:	umlungu consulting (pty) ltd
	Author:		Alan Benington
	Started:	2008-06-14
	Status:		release	
	Version:	2.0.0
	Build:		20080614
	License:	GNU General Public License
	======================================================================= -->

<!--=======================================================================
	Development Notes:
	==================
	20080614:	Initial release.
	=======================================================================	-->

<xsl:stylesheet xmlns:xsl='http://www.w3.org/1999/XSL/Transform' version='1.0'>

<!--=======================
	Match templates section
	======================= -->
	<!-- top menu templates -->
	<xsl:template match="website/menus">
		<ul id="top-mnu">
			<xsl:attribute name="onmouseover">javascript:x_page.CheckDrop(false, this)</xsl:attribute>
			<xsl:attribute name="onmouseout">javascript:x_page.CheckDrop(true, this)</xsl:attribute>
			<xsl:apply-templates select="menu[not(@display) or @display!='no']"/>
		</ul>
	</xsl:template>
	
	<xsl:template match="website/menus/menu">
		<li class="top">
			<xsl:if test="position()=last()">
				<xsl:attribute name="class"><xsl:text>top last</xsl:text></xsl:attribute>
			</xsl:if>
			<xsl:value-of select="@label"/>
			<ul>
				<xsl:apply-templates select="menuitem"/>
			</ul>
		</li>
	</xsl:template>
	
	<xsl:template match="website/menus/menu//menuitem">
		<li class="mnu-lft">
			<xsl:choose>
				<xsl:when test="menuitem">
					<xsl:choose>
						<xsl:when test="@link">
							<a>
								<xsl:attribute name="href">
									<xsl:value-of select="$BaseLinkUrl"/>
									<xsl:value-of select="@link"/>
								</xsl:attribute>
								<xsl:value-of select="@label"/>
							</a>
						</xsl:when>
						<xsl:otherwise>
							<xsl:value-of select="@label"/>
						</xsl:otherwise>
					</xsl:choose>
					<!-- 
					<ul id="menu-lft">
						<xsl:apply-templates select="menuitem"/>
					</ul>
					-->
				</xsl:when>
				<xsl:otherwise>
						<xsl:if test="@link=$link">
							<xsl:attribute name="class">mnu-lft-cur</xsl:attribute>
						</xsl:if>
						<a>
							<xsl:attribute name="href">
								<xsl:value-of select="$BaseLinkUrl"/>
								<xsl:value-of select="@link"/>
							</xsl:attribute>
							<xsl:value-of select="@label"/>
						</a>
				</xsl:otherwise>
			</xsl:choose>
		</li>
	</xsl:template>
	<!---->
	<xsl:template match="website/menus/menu/menuitem/menuitem">
		<li class="mnu-lft">
			<xsl:if test="@link=$link">
				<xsl:attribute name="class">mnu-lft-cur</xsl:attribute>
			</xsl:if>
			<a>
				<xsl:attribute name="href">
					<xsl:value-of select="$BaseLinkUrl"/>
					<xsl:value-of select="@link"/>
				</xsl:attribute>
				<xsl:value-of select="@label"/>
			</a>
		</li>
	</xsl:template>
	
	<!-- left hand menu templates -->
	<xsl:template match="website/menus/menu" mode="left">
		<h3><xsl:value-of select="@label"/></h3>
		<ul id="menu-lft">
			<xsl:apply-templates select="menuitem"/>
		</ul>
	</xsl:template>
	
	<!-- footer menu templates -->
	<xsl:template match="website/menus/footmenu">
		<div id="footer">
			<xsl:if test="$device=''">
				<div class="foot">
					<span>SEARCH</span>
					<form method="post" name="SearchForm">
						<xsl:attribute name="action">
							<xsl:value-of select="$BaseLinkUrl"/>
							<xsl:text>site_search</xsl:text>
						</xsl:attribute>
						<input type="text" id="SearchText" name="SearchText"></input>
					</form>
				</div>
				<xsl:apply-templates select="menu"/>
			</xsl:if>
			<xsl:apply-templates select="copyright"/>
		</div>
	</xsl:template>
	
	<xsl:template match="website/menus/footmenu/menu">
		<div class="foot">
			<xsl:apply-templates/>
		</div>
	</xsl:template>
	
	<xsl:template match="website/menus/footmenu/menu/menuitem">
		<xsl:if test="position()!=1">
			<span class="sep">|</span>
		</xsl:if>
		<span>
			<a>
				<xsl:attribute name="href">
					<xsl:value-of select="$BaseLinkUrl"/>
					<xsl:value-of select="@link"/>							
				</xsl:attribute>
				<xsl:value-of select="@label"/>
			</a>
		</span>
	</xsl:template>
	
	<xsl:template match="website/menus/footmenu/copyright">
		<div id="copyright"><xsl:apply-templates/></div>
	</xsl:template>
	
	<!-- related links templates -->
	<xsl:template match="link/links">
		<xsl:call-template name="RelatedLinks"/>
	</xsl:template>	
	
	<xsl:template match="link/links/link">
		<xsl:variable name="thisname" select="@name"/>
		<li>
			<a><xsl:attribute name="href">
					<xsl:choose>
						<xsl:when test="@www">
							<xsl:text>javascript:x_page.Popup('</xsl:text>
							<xsl:text>http://</xsl:text><xsl:value-of select="@www"/>
							<xsl:text>')</xsl:text>
						</xsl:when>
						<xsl:otherwise>
								<xsl:value-of select="$BaseLinkUrl"/>
								<xsl:value-of select="@name"/>							
						</xsl:otherwise>
					</xsl:choose>
				</xsl:attribute>
				<xsl:choose>
					<xsl:when test="text()"><xsl:value-of select="."/></xsl:when>
					<xsl:otherwise><xsl:value-of select="//website/links/link[@name=$thisname]/prompt"/></xsl:otherwise>
				</xsl:choose>
			</a>
		</li>
	</xsl:template>	
	
	<xsl:template match="link/links/popup">
		<input type="Button">
			<xsl:attribute name="value">
				<xsl:choose>
					<xsl:when test="."><xsl:value-of select="."/></xsl:when>
					<xsl:otherwise><xsl:text>Open</xsl:text></xsl:otherwise>
				</xsl:choose>
			</xsl:attribute>
			<xsl:attribute name="onclick">
				<xsl:call-template name="popup-action"/>
			</xsl:attribute>
		</input>
	</xsl:template>	
	
	<xsl:template name="popup-action">
		<xsl:param name="frmname"/>
		<xsl:choose>
			<xsl:when test="@url">
				<xsl:text>javascript:x_page.Popup('</xsl:text>
				<xsl:value-of select="@url"/>
				<xsl:text>')</xsl:text>
			</xsl:when>
			<xsl:when test="@width">
				<xsl:choose>
					<xsl:when test="@root='yes'"><xsl:text>javascript:x_page.PopRoot('</xsl:text></xsl:when>
					<xsl:when test="@scroll='yes' and @post='yes'"><xsl:text>javascript:x_page.PopScrollPost('</xsl:text></xsl:when>
					<xsl:when test="@scroll='yes'"><xsl:text>javascript:x_page.PopScroll('</xsl:text></xsl:when>
					<xsl:otherwise><xsl:text>javascript:x_page.Pop('</xsl:text></xsl:otherwise>
				</xsl:choose>
				<xsl:choose>
					<xsl:when test="@www"><xsl:text>http://</xsl:text><xsl:value-of select="@www"/></xsl:when>
					<xsl:otherwise><xsl:value-of select="@src"/></xsl:otherwise>
				</xsl:choose>
				<xsl:text>', '</xsl:text><xsl:value-of select="@width"/>
				<xsl:text>', '</xsl:text><xsl:value-of select="@height"/>
				<xsl:if test="@post='yes'">
					<xsl:text>', '</xsl:text><xsl:value-of select="@name"/>
				</xsl:if>
				<xsl:text>')</xsl:text>
			</xsl:when>
			<xsl:otherwise>
				<xsl:choose>
					<xsl:when test="@post='yes'"><xsl:text>javascript:x_page.PopCoverPost('</xsl:text></xsl:when>
					<xsl:otherwise><xsl:text>javascript:x_page.PopCover('</xsl:text></xsl:otherwise>
				</xsl:choose>
				<xsl:choose>
					<xsl:when test="@www"><xsl:text>http://</xsl:text><xsl:value-of select="@www"/></xsl:when>
					<xsl:otherwise><xsl:value-of select="@src"/></xsl:otherwise>
				</xsl:choose>
				<xsl:if test="@post='yes'">
					<xsl:text>', '</xsl:text><xsl:value-of select="@name"/>
				</xsl:if>
				<xsl:text>')</xsl:text>
			</xsl:otherwise>
		</xsl:choose>
	</xsl:template>	
	
	<!-- home page links templates -->
	<xsl:template match="page/links">
		<xsl:comment>home links:</xsl:comment>
		<xsl:if test="link">
			<ul id="home-link">
				<xsl:apply-templates select="link"/>
			</ul>
		</xsl:if>
	</xsl:template>	
	
	<xsl:template match="page/links/link">
		<xsl:variable name="thisname" select="@name"/>
		<li>
			<a>
				<xsl:attribute name="href">
					<xsl:value-of select="$BaseLinkUrl"/>
					<xsl:value-of select="@name"/>							
				</xsl:attribute>
				<xsl:value-of select="."/>
			</a>
		</li>
	</xsl:template>	
	
	<xsl:template match="link">
		<xsl:variable name="thisname" select="@name"/>
			<a><xsl:attribute name="href">
					<xsl:value-of select="$BaseLinkUrl"/>
					<xsl:value-of select="@name"/>							
				</xsl:attribute>
				<xsl:choose>
					<xsl:when test="text()"><xsl:value-of select="."/></xsl:when>
					<xsl:otherwise><xsl:value-of select="//website/links/link[@name=$thisname]/prompt"/></xsl:otherwise>
				</xsl:choose>
			</a>
	</xsl:template>	
	
<!--=======================
	Named templates section
	======================= -->
	
	<xsl:template name="RelatedLinks">
		<xsl:apply-templates select="popup"/>
		<xsl:if test="link">
			<ul id="rellink">
				<xsl:apply-templates select="link"/>
			</ul>
		</xsl:if>
	</xsl:template>
	
	<xsl:template name="simple-link">
		<xsl:param name="link"/>
		<xsl:param name="hash"/>
		<xsl:value-of select='$base'/><xsl:value-of select='$BaseLinkUrl'/><xsl:value-of select='$link'/>
		<xsl:if test="$hash"><xsl:text>#</xsl:text><xsl:value-of select="$hash"/></xsl:if>
	</xsl:template>
	
	<!-- like simple link, but pops-up-->
	<xsl:template name="popup-link">
		<xsl:param name="link"/>
		<xsl:text>javascript:x_page.PopUp('</xsl:text><xsl:value-of select='$base'/><xsl:value-of select='$BaseLinkUrl'/><xsl:value-of select='$link'/>
		<xsl:text>');</xsl:text>
	</xsl:template>
	
	<!-- skeleton: needs definition -->
	<xsl:template name="url">
		<xsl:param name="link"/>
	</xsl:template>
	
</xsl:stylesheet>
